TX-0 COMPUTER
MASSACHUSETTS INSTITUTE OF TECHNOLOGY
CAMBRIDGE 39, MASSACHUSETTS

M-5001-28

PROGRAMMING FOR THE MAGNETIC TAPE SYSTEM

July 7, 1960

PROGRAMMING FOR THE MAGNETIC TAPE SYSTEM

Initially, the magnetic tape system will contain a Potter Model 906II tape handler. The control electronics will allow expansion to three tape units, and a photoelectric tape reader. The recording system used is compatible with IBM magnetic tape systems 727, 729I, 729II, and 729IV.

Information on the Tape

Information is recorded on tape by writing over any previous information. Only ones or zeros are written on the tape. Six bits plus a parity bit are written laterally across the tape to form a character. Written characters, of course, can then be read. Three characters form a TX-0 word. Two-hundred characters are stored per linear inch of tape, and the tape moves at a speed of 75 inches/second while tape is being written, read, or backspaced (reading backwards).

Groups of word(s) form records. The 3/4" space between records is known as an end-of-record gap, eor. The layout of the tape is shown in Fig. 1. Reflective strips placed at each end of the tape are sensed to indicate the load point, ldp, and end-of-tape, eot.

Select commands, opr ~~40000~~ 4000, are used to specify the unit and the operation to be effected. Three units and five modes may be selected. Bits 16,17 of the opr 40000 instruction determine the unit number and bits 13,14,15 specify the operating mode. A one in bit 2 of a select instruction will cause the AC to be cleared. Bits 3 and 9-13 are unused, but should be zero to allow for possible future expansion.

Writing

The select write mode, wrm+n, is opr ~~40000~~ 4000+14+n, where n=1,2,or 3 specifies the unit number. Writing is done by giving the wrm n instruction

followed by a copy order for each word to be written. The copy order, cpy, is opr 20000, and may be used with micro orders specified by bits 2,3,9-17. Words may be written (copied) on the tape 4.6 ms after the wrm n is given. The first cpy will be temporarily halted until the 4.6 ms delay has terminated. If no copy is given during the initial 4.6 ms interval, the unit begins to disconnect.

The cpy writes the contents of the live register, LR, on tape. A cpy momentarily halts the machine, and is executed 198 μs after a previous cpy. The LR is used for 132 μs after a cpy is given, by the tape system, and therefore should not be destroyed by a program during this time. The contents of the LR are changed during the writing. The command, cpy+alr-opr, may be used to make sure the LR is free, and avoid program timing, i.e., the cpy is automatically held up until a previous cpy is complete, and the alr mic -command is executed just before anything is written. The instruction sequence, cpy followed by alr, would not correctly write the contents of the ac on tape. The instruction alr, followed by a cpy, would not operate properly unless 132 μs had elapsed since the previous cpy.

When the last cpy for a record has been executed, i.e., a cpy has not been given for 198 μs, the unit continues on to write the end-of-record gap and is then disconnected (deselected). Deselection requires 5.75 ms and calculations may proceed normally during this interval. If a cpy is given and no unit is selected, the cpy will be executed as a no-operation instruction and the Read-Write flip-flop/flag (rwc) will be set.

The following nonsense writing program illustrates various timing conditions:



FIGURE 1 - LAYOUT OF CHANNELS, CHARACTERS, WORDS, AND RECORDS OF INFORMATION

```
| Program for writing nonsense
        cpy=opr 20000
        wrm 1       | unit 1 begins writing, normal programming may
                    | be done for next 4.6 ms.
        llr a
        cla+20000   | this cpy instruction waits the 4.6 ms., then the
                    | contents of register a are written on tape

        add b
        ala 20000   | this cpy executed 198 microseconds after previous
                    | cpy, the contents of b are written on tape


        add (-7+1
        add (1
        trn .-1     | 138 microsec. delay loop


        cla
        add c
        ala         | lr free at this time.
        cpy         | cpy executed 198 microsec. after previous cpy, the
                    | contents of c begin to be written on tape


        add d
        cpy         | contents of lr, i.e., the scrambled data left over
                    | from the previous cpy begins to be written on tape,
        ala         | this instruction destroys the contents of the lr
                    | before the scrambled data can be completely written, thus,
                    | the contents of c scrambled, and the contents of d
                    | are written on tape.


        add (-14+1
        add (1
        trn .-1     | 216 microsecond delay


        cpy         | more than 198 microseconds have elapsed from time of
                    | previous cpy, this cpy is held up for 5.6 ms (or
                    | until the unit is disconnected).  A cpy with no
                    | unit selected sets the rwc flag and is executed as
                    | a no operation instruction.

        wrm 1       | write an extra long gap for end of record
        wrm 1       | instruction is temporarily halted.  Deselection
                    | from previous wrm 1 requires (4.6+5.75)ms.  live
                    | register is free and if another select order is now
                    | given, it will be halter for (5.75+4.6)ms.


        hlt         | end of program
```

The preceeding program writes four words on tape; C(a), C(b), C(c), and C(c) jumbled with C(d). Two extra end-of-record gaps are written after the four-word record. The rwc flag is set.

A fair dinkum writing loop is presented in Appendix I.

Reading and Spacing Forward

The select read mode, rdm+n is given by the command opr 40000+4+n. The rdm command starts the tape moving forward ready to transmit information to the IR. About (4.5+4) ms will elapse before the first word actually appears in the IR. Thus, the IR may be safely used for about 7 ms following an rdm order. Execution of the rdm order guarantees that the tape will move forward, over the record, to the following end-of-record and be deselected. The end-of-record flag, eor, will be set when the tape has moved over the record.

The copy instruction is used to synchronize information entering the IR. A cpy is temporarily held up until a complete word is assembled in the IR. Therefore, the IR should not be disturbed while the word, composed of three characters, is being assembled. If a copy has not been given by the time a complete word has entered the IR, the rwc flag is set -- signifying that the program has missed information coming from the tape. Copy instructions in this case should be given not more than 170 μs apart. The IR may be used for computation during the first 40 μs following a cpy.

Forward spacing may be effected by giving an rdm n followed by no cpys. The eor flag is set when the tape has been moved to the end-of-record. During forward spacing (no cpy has been given) no information enters the IR, and normal computation may be carried out during the entire time of selection.

During reading, the parity bit is checked and the pc flag set if a character is in error. Forward spacing does not check the parity bit.

A reading loop is presented in Appendix I.

Backspacing

The select backspace order, bsm n=opr 40000+n, spaces the tape backwards one record. The tape is positioned to the previous end of record, and the eor flag is set. If the tape is at the load point, the bsm n order is executed as a no operation instruction, and the rwc flag is set.

Rewinding

The select rewind order, rem n=opr 40000+10+n, moves the tape of unit n to the load point. The order to start the rewinding is executed in 12 μs, and control is deselected. A select order affecting a unit that is rewinding is held up until the rewind is complete.

Unit Availability

Under certain conditions, a unit will appear unavailable to the program. These conditions are:

1. Unit is rewinding
2. Tape improperly loaded
3. Cover door open
4. Unit power off
5. Unit overloaded
6. Unit being controlled by manual switches.

An order selecting an unavailable unit will beheld up until the unit becomes available.

If a unit becomes unavailable during writing, reading, or backspacing, it will be stopped immediately, deselected, and the rwc flag will be set. Thus, the select order will probably be executed incorrectly.

When the tape comes to the reflective marker at the end-of-tape, the eot flag will be set. A Tape driven past the marked end-of-tape will eventually be pulled from the supply reel and the tape will become improperly loaded, causing the unit to become unavailable.

Program Flags for Tape

The order read program flags, rpf=opr 6000, forms the inclusive or between the flag flip-flops and the MBR leaving the result in the MBR. The flags may be placed in the AC or IR for testing purposes. The order set program flags, spf=opr 7000, places the MBR contents in the program flag flip-flops. The following table summarizes the conditions for which flags are set, and the bit positions in the MBR.

MBR bit	Flag	Conditions for Setting Flag			
		General	cpy	rdm n	bsm n
4	rwc	when an operating unit becomes anavailable (unit will also be deselected)	when nothing is selected (cpy executed as a no-operation)	if a complete word enters lr and a copy is not given in time (the first cpy will not set rwc)	when tape is at ldp (bsm executed as a no-operation)
5	eor	------	------	when tape comes to next end of record	when tape comes to end of previous record gap
6	pc	------	------	set if parity bit is in error (not used for forward spacing)	------
7	eot	when reflective strip is sensed at the end of tape	------	------	------

SUMMARY OF FLAGS

APPENDIX I

PROGRAMS FOR WRITING AND READING

Program to write an n word record stored beginning in register buf.

```
        xx=hlt
buf,    0                       | first word of record to be written
buf+n-1|          0             | last word of record written

fir,    spf                     | clear flags
        wrm 1                   | start the writing
        llr (add buf
        slr .+2

        cla                     | start of write loop
        xx                      | becomes add buf, add Buf+1,...add buf+n
        cpy+ala-opr             | buf,buf+1,...buf+n-1 goes on tape
        add (1
        add .-3
        sto .-4
        add (-add-buf-n+1       | test word for n times
        trn .-7                 | loop end

        rpf+140020              | flags to ac for checking
        cyl
        cyl
        cyl
        cyl
        trn er1                 | er1 for rwc, therefore unit not operating ok.
        cyl
        cyl
        cyl
        trn er2                 | er2 for eot
        wrm 1                   | write an extra long end of record gap
end,    hlt                     | end of program stop
```

```
| Program to read n words from tape to storage beginning in register buf


sec,    spf                     | clear program flags
        rdm 1                   | begin looking for words
        llr (slr buf
        slr .+2


        cpy+cla-opr             | beginning of loop for reading
        xx                      | slr buf,slr buf+1,...slr buf+n-1
        add (1
        add .-2
        sto .-3
        add (-slr-buf-n+1
        trn .-6                 | loop end


        rpf+140020              | begin to check program flags
        cyl
        cyl
        cyl
        cyl


        trn er1                 | er1 for rwc, something wrong.
        cyl
        trn .+2
        tra er3                 | er3 for not eor, should be at end of record
        cyl
        trn er4                 | er4 for parity, a word read incorrectly
        cyl
        trn er2                 | er2 for eot, end of tape


las,    hlt                     | halt at program end
```

	Available time before first cpy must be given	Time between cpy's	Useable time for IR between cpy's	Time after eor for unit to deselect	Time for manipulating the record	Flags that may be set
wrm n opr 40014	4.6 ms (330)	198 μs (14 1/2) (any longer time deselects unit)	66 μs (4) (not available until 132 μs after a cpy)	5.75 ms (400)	(4.6+5.75+.198x no. of words in record) ms.	rwc (if unit becomes unavailable) eot
rdm n opr 40040	8.6 ms (550)	198 μs -(13) (any longer time misses information)	66 μs (4) (available just after cpy completed)	1.75 ms (125)	(8.6+1.75+.198x no. of words in record) ms.	rwc (if unit becomes unavailable or a cpy is not given in time) pc,eot,eor
rdm n (with no cpy)		Requires same times as rdm n above, but IR is not used				eot,eor,rwc (if unit becomes unavailable)
opr 40000 bsm n (cpy not used)			bsm n does not use lr		(4.6+5.75+.198x no. of words in record) ms.	rwc(unit becomes unavailable, or tape at ldp) eor
nothing selected	------	------	------	------	------	rwc - if a cpy is given

APPENDIX II - SUMMARY OF COMMANDS

NOTE: (x) is no. of 12 μs commands that may be safely given in the interval.